DAΦNE TECHNICAL NOTE

INFN - LNF, Accelerator Division

Frascati, September 3, 1992

Note: **LC-5**

DAΦNE-LINAC BEAMS EMITTANCE MEASUREMENT DESIGN

F. Sannibale

In the first part of this note the application of the well known 'scanning quad' method [1] to the measurement of the DAΦNE-LINAC beam transverse emittances is discussed. This method allows also the evaluation of the beam optical functions (Twiss parameters).

In the last part there is a short description of the software performing the data analysis.

1 - The 'scanning quad' method

1.1 *General description*

The emittance and the Twiss parameters values are derived indirectly by beam profile measurements. The apparatus set-up is schematically shown in Fig. 1. A set of beam dimensions for different strengths of the quadrupole (one of the DAΦNE transferline) is registered and analyzed in order to derive emittance and Twiss parameters. In § 1.3 it will be described how this analysis is performed.



Figure 1 : measurement set-up block schematics.

The method can be used for beams with the following features:

- *linear behavior*: the beam transport can be represented, to good approximation, by the first order formalism.

- *emittance dominance*: the beam particles dynamics in a drift space has to be dominated by the emittance action. In particular the effects of the space charge have to be negligible. This condition is fulfilled when in the beam envelope equation the ratio between the space charge and the emittance terms is << 1 [2]:

$$\Re = \frac{I R^2}{\varepsilon^2 I_0 (\beta\gamma)^3} \ll 1$$

with

$$I_0 = \frac{2\pi\, \varepsilon_0 m_0 c^3}{e}$$

where I is the beam current, R the typical beam half envelope, ε the beam emittance, β and γ the Lorentz parameters, m_0 and e the electron rest mass and charge, c the light speed and ε_0 the permittivity of the free space.

The main characteristics of the DAΦNE-LINAC beams are summarized in Table 1.

The transferline elements are designed in order to have very low values of multipolar field terms [3]. This means that the first order transport formalism is a good model.

Finally the values we obtain for the ratio $\Re$:

5×10^{-9} for the positron beam
2×10^{-6} for the electron beam

indicate that the method is surely applicable.

Table 1: DAΦNE-LINAC beams design parameters

	Positrons	Electrons
Energy	510 MeV	510÷800 MeV
Current	40 mA	200 mA
Pulse length	10 ns	10 ns
Geometric emittance (@ 510 MeV)	10^{-5} m rad	10^{-6} m rad
Typical envelope (half width)	1 cm	1 cm

1.2 *Measurement set-up*

Figure 2 indicates as the experimental set-up will be implemented on the DAΦNE-LINAC. It's composed by the beam profile monitor (BPFTM01) on the beam line and the quadrupole (QUATM03). The beam profile monitor will be of the same kind of the one used at CERN, the so-called 'Chromox CERN type 6', produced by Morgan Matroc Ltd., Anderman Division, East Molesey, Surrey, U.K. This is a chrome doped alumina screen which has been extensively tested at CERN by C. D. Johnson [4]. Its most important feature is the linearity of the response for current densities of the same order we will have in our Linac. It has been already successfully used for emittance measurements in ELETTRA [5].



Figure 2 : measurement set-up general lay-out.

The distance between the beam profile monitor and the upstream quadrupole (2.9 m) has been optimized, through the LEDA code [6], in order to have on the monitor (in the range of the available quadrupole strengths) an envelope minimum for both the beams. This feature as it will be shown later is very important for the method accuracy. To commute from one beam measurement to the other one it is necessary to change the set of the three quadrupoles after the Linac end (QUATM01,2,3). To make easier the measurement and the transport of the beam a wall current monitor (WCMTM01) and H&V correctors (CHVTM01) have been included in the set-up.

1.3 *Theory*

The scanning quad method uses the quadrupole thin lens description. This approximation is valid when [7]:

$$k^2 L^2 \ll 1 \tag{1}$$

where L and k^2 are the length and the focusing strength of the quadrupole. The DAΦNE-Transferline quadrupoles length is 20 cm, this means

$$k^2 \ll 25\ m^{-2}$$

The thin quad matrix in the focusing case is:

$$\begin{pmatrix} 1 & L/2 \\ 0 & 1 \end{pmatrix}\begin{pmatrix} 1 & 0 \\ A & 1 \end{pmatrix}\begin{pmatrix} 1 & L/2 \\ 0 & 1 \end{pmatrix} \tag{2}$$

where

$$A = - k \sin kL \tag{3}$$

The complete matrix from the center of the quad to the end of the following drift is:

$$\begin{pmatrix} 1 & 0 \\ A & 1 \end{pmatrix}\begin{pmatrix} 1 & D \\ 0 & 1 \end{pmatrix} = \begin{pmatrix} 1+DA & D \\ A & 1 \end{pmatrix} \tag{4}$$

where

$$D = L/2 + d \tag{5}$$

with d the length of the drift between the quadrupole end face and the beam profile monitor.

From (4)

$$x = (1 + DA)x_0 + Dx_0'$$

where x and x' are the coordinates of a beam particle in the analyzed phase plane.

The beam envelope r.m.s. value is given by the square root of $\langle x^2 \rangle$:

$$\langle x^2 \rangle = (1 + DA)^2 \langle x_0^2 \rangle + D^2 \langle x_0'^2 \rangle + 2D(1 + DA)\langle x_0 x_0' \rangle \tag{6}$$

The (6) is a parabola with respect to the variable A which assumes, when $A = A_m$, its minimum value $<x^2>_m$,. It's easy to check that:

$$A_m = -\frac{1}{D} - \frac{<x_0\, x_0'>}{<x_0^2>} \tag{7}$$

$$<x^2>_m = \frac{D^2}{<x_0^2>}\left(<x_0^2><x_0'^2> - <x_0\, x_0'>^2\right)$$

from the sigma formalism we know that:

$$\sigma_{11} = <x^2> \qquad \sigma_{22} = <x_0'^2> \qquad \sigma_{12} = <x_0\, x_0'> \tag{8a}$$

$$\varepsilon_{r\,m\,s} = \sqrt{\sigma_{11}\sigma_{22} - \sigma_{12}^2} \tag{8b}$$

with ε_{rms} the root mean square emittance. Therefore we can write:

$$<x^2>_m = \frac{D^2}{<x_0^2>}\ \varepsilon_{r\,m\,s}^2 \tag{9}$$

Now using only the expression (6) we obtain after a little algebra:

$$<x^2> - <x^2>_m = D^2\,(A - A_m)<x_0^2>\left(A + A_m + \frac{2}{D} + 2\,\frac{<x_0\, x_0'>}{<x_0^2>}\right)$$

and by making use of (7)

$$<x^2> - <x^2>_m = D^2\,(A - A_m)<x_0^2>\,(A + A_m - 2A_m)$$

or

$$<x_0^2> = \frac{<x^2> - <x^2>_m}{D^2\,(A - A_m)^2} \tag{10}$$

By using (9) in (10) we obtain:

$$\varepsilon_{r\,m\,s} = \frac{\sqrt{<x^2>_m\left(<x^2> - <x^2>_m\right)}}{D^2\,|A - A_m|} \tag{11}$$

and by (7) and (10)

$$<x_0 x_0'> = -\,\frac{1 + A_m D}{D}\ \frac{<x^2> - <x^2>_m}{D^2\,(A - A_m)^2} \tag{12}$$

Finally from (8), (10), (11) and (12) we get:

$$<x_0'^2> = \frac{<x^2>_m}{D^2} + \frac{(1 + A_m D)^2}{D^2}\ \frac{<x^2> - <x^2>_m}{D^2\,(A - A_m)^2} \tag{13}$$

The relations (10), (11), (12) and (13) are the ones we are looking for. By means of beam profile measurements we can obtain $<x^2>$ [1]. In fact, if R indicates the measured r.m.s. value of the beam half envelope then:

$$<x^2> = R^2 \tag{14}$$

Making several profile measurements for different values of the quadrupole strength and fitting the data by a parabola it is possible to obtain A_m and $<x^2>_m$.

At this point using the relations (10), (11), (12), (13) and the well known ones for the optical functions (Twiss parameters):

$$\beta = \frac{<x_0^2>}{\varepsilon_{rms}} \qquad \gamma = \frac{<x_0'^2>}{\varepsilon_{rms}} \qquad \alpha = -\frac{<x_0 x_0'>}{\varepsilon_{rms}} \tag{15}$$

we can calculate a set of values for the emittance and the optical functions: one for each different beam profile measurement. Computing the mean value and the standard deviation of this set we finally obtain the desired results with their statistical indeterminacy.

The obtained optical functions are those the beam has at the quad center. To obtain them at the quad beginning is then necessary to apply to the 'vector' (β, α, γ) the inverse matrix of a drift of length L/2 (because the thin lens approximation, see (2)):

$$\begin{pmatrix} 1 & L & \frac{L^2}{4} \\ 0 & 1 & \frac{L}{2} \\ 0 & 0 & 1 \end{pmatrix}$$

1.3 *Considerations on the method accuracy*

The accuracy of the method strongly depends on the goodness of the parabolic fit. To obtain this requirement the following 'rules' have to be satisfied:

- Several different profile measurements have to be done.
- Very accurate profile measurements are necessary.
- Goodness of the thin lens model. See expression (1).
- The set of the profile measurements has to include quadrupole strength values on the right and on the left of the value concerning the minimum envelope.

---

[1] These profile measurements will be done by a fluorescent screen. To know the r.m.s.value of the beam envelope we have two possibilities. The first one is to exactly calculate the value using the intensity information of each pixel of the digitized image. The other one is to assume a beam gaussian profile and to calculate the r.m.s. envelope by fitting the data. The first method is more accurate but needs a very long computing time. The second one is very quick but if the beam is not gaussian it could introduce large errors in the measured values. The final decision about the fairest method will be taken observing the first profile measurements.

Furthermore it is preferable to avoid the use of the expressions (10), (11), (12) and (13) when $A \approx A_m$. The reason is because in all them both the numerator and the denominator vanish when $A = A_m$. In these conditions a very high precision in calculation is needed to avoid large errors.

2- The 'Emittance Routine'

2.1 *Main features*

The *Emittance Routine* is written in FORTRAN 77 and applies the theory described in the previous chapter. It requires as input the measured beam profile values and indeterminacies and gives as output the values and indeterminacies of emittance and beam optical functions at the quadrupole beginning. Informations about the fit and its goodness are also provided (fit values and indeterminacies, chisquare variable, degrees of freedom). In Appendix the routine list, input and output files are given.

2.2 *Routine test*

For testing purposes, a set of beam profile values has been simulated by the means of the first order matrix code LEDA. The values so obtained have been used in the Emittance Routine and the results have been compared with the known ones. Very good agreement, within few percent, has been obtained in this ideal case.

REFERENCES

[1] R. Miller. Method's father. Private communications.

[2] F. Sannibale *Progetto realizzazione ed uso di un misuratore di emittanza....* Thesis. Universita' degli Studi di Roma "La Sapienza".

[3] C. Biscari, F. Sannibale *Tolerances and corrections scheme in DAΦNE transfer Lines* DAΦNE Technical Note I-10 (June 18, 1992).

[4] C. D. Johnson *The Development and Use of Alumina Ceramic Fluorescent Screens* CERN PS/90-42(AR).

[5] J.-C. Denard et al. *Beam Diagnostic of the ELETTRA Injector* EPAC 92 Berlin 24-28 march 1992.

[6] J. B. Murphy, G. Vignola, LEDA code (unpublished).

[7] K. G. Steffen *High Energy Beam Optics* Interscience Publishers (John Wiley & Sons).

APPENDIX

EMITTANCE.FOR

```
c          'EMITTANCE'  by Fernando Sannibale (started: August 6,92)
c
c
c        ========================MAIN================================
c
         IMPLICIT REAL*8 (A-H,O-Z)
         INCLUDE 'Comblock.for'
c
         CALL Input
         CALL LMSparabola
         CALL Twiss
         CALL Topoutput
         CALL Output
c
         STOP
         END
c
c        ===========================================================
c        ********************ROUTINES*******************************
c        ===========================================================
c
c        ###########################################################
         SUBROUTINE Input
c
         IMPLICIT REAL*8 (A-H,O-Z)
         INCLUDE 'Comblock.for'
c
c        Parameters:
c
         pem=9.1095D-31
         pec=1.6022D-19
         pc=2.9979D+8
         pi=3.14159265359
c
1        READ (10,1001) title
2        READ (10,1001) blob
3        READ (10,1001) blob
4        READ (10,*) energy,current,bunchlen
5        READ (10,1001) blob
6        READ (10,1001) blob
7        READ (10,1001) blob
8        READ (10,*) quadlen,driftlen
9        READ (10,1001) blob
10       READ (10,1001) blob
11       READ (10,1001) blob
12       READ (10,*) iraveflag
13       READ (10,1001) blob
14       READ (10,1001) blob
15       READ (10,1001) blob
16       READ (10,*) npoints
17       READ (10,1001) blob
18       READ (10,1001) blob
19       READ (10,1001) blob
20       READ (10,1001) blob
21       READ (10,1001) blob
22       READ (10,1001) blob
c
         gamma=1.+1.957*energy
         akmin=1.d19
         akmax=-1.d19
         sigmin=1.d19
         sigmax=0.
c
         DO 100,i=1,npoints
             READ (10,*) itrash,akappa2(i),sigma(i),dsigma(i)
             akappa(i)=DSQRT(akappa2(i))
             akappa(i)=akappa(i)*DSIN(akappa(i)*quadlen)
             akappa(i)=-akappa(i)*1.d-3
             IF (dsigma(i).EQ.0) THEN
                 izerodsigflag=1
                 dsigma(i)=1.
             END IF
c
             IF (akmin.GT.akappa(i)) akmin=akappa(i)
             IF (akmax.LT.akappa(i)) akmax=akappa(i)
             IF (sigmin.GT.sigma(i)) sigmin=sigma(i)
             IF (sigmax.LT.sigma(i)) sigmax=sigma(i)
c
100      CONTINUE
c
```

```
c
c       -------
        RETURN
1001    FORMAT (A80)
        END
c
c       ####################################################
        SUBROUTINE LMSparabola
c
        IMPLICIT REAL*8 (A-H,O-Z)
        REAL*8 hprov(3)
        INCLUDE 'Comblock.for'
c
        DO 201,ii=1,3
           DO 202,jj=1,3
               h(ii,jj)=0.
202     CONTINUE
201     CONTINUE
        DO 100,i=1,npoints
c
            derr=2.*sigma(i)*dsigma(i)
            h(1,1)=h(1,1)+akappa(i)**2.*akappa(i)**2./derr**2.
            h(1,2)=h(1,2)+akappa(i)**2.*akappa(i)/derr**2.
            h(1,3)=h(1,3)+akappa(i)**2./derr**2.
            h(2,1)=h(1,2)
            h(2,2)=h(1,3)
            h(2,3)=h(2,3)+akappa(i)/derr**2.
            h(3,1)=h(1,3)
            h(3,2)=h(2,3)
            h(3,3)=h(3,3)+1./derr**2.
c
            a(1)=a(1)+sigma(i)**2.*akappa(i)**2./derr**2.
            a(2)=a(2)+sigma(i)**2.*akappa(i)/derr**2.
            a(3)=a(3)+sigma(i)**2./derr**2.
c
100     CONTINUE
c
        CALL Determinant(h,hdet)
        DO 200,i=1,3
            DO 300,j=1,3
                hprov(j)=h(j,i)
                h(j,i)=a(j)
300         CONTINUE
            CALL Determinant(h,prt)
            par(i)=prt/hdet
            DO 400,j=1,3
                h(j,i)=hprov(j)
400         CONTINUE
c
            ii1=i+1
            IF (ii1.EQ.4) ii1=1
            ii2 =i+2
            IF (ii2.EQ.4) ii2=1
            IF (ii2.EQ.5) ii2=2
            dpar(i)=h(ii1,ii1)*h(ii2,ii2)-h(ii1,ii2)*h(ii2,ii1)
            dpar(i)=DSQRT(DABS(dpar(i)/hdet))
200     CONTINUE
c
        chisquare=0.
        DO 500 i=1,npoints
            derr=2.*sigma(i)*dsigma(i)
            chih=sigma(i)**2-par(1)*akappa(i)**2-par(2)*akappa(i)-par(3)
            chisquare=chisquare+(chih/derr)**2
500     CONTINUE
c
        ndegfree=npoints-3
c       -------
        RETURN
        END
c
c       ----------------------------------------------------
        SUBROUTINE Determinant(aa,detval)
c
        IMPLICIT REAL*8 (A-H,O-Z)
        REAL*8 aa(3,3),ia(3)
c
        detval=0.
        ia(1)=0
        ia(2)=1
        ia(3)=2
```

```
      DO 100,i=1,3
          adet1=1.
          adet2=1.
          DO 200,j=1,3
              ia(j)=ia(j)+1
              IF (ia(j).EQ.4) ia(j)=1
              adet1=adet1*aa(j,ia(j))
              adet2=adet2*aa(j,4-ia(j))
200       CONTINUE
          detval=detval+adet1-adet2
100   CONTINUE
c     --------
c
      RETURN
      END
c
c     ###########################################################
      SUBROUTINE Twiss
c
      IMPLICIT REAL*8 (A-H,O-Z)
      INCLUDE 'Comblock.for'
c
c
      akm=-par(2)*1.d-9/2./par(1)/1.d-12
      sig2min=par(3)*1.d-6+akm*par(2)*1.d-9/2.
      qdl=driftlen+quadlen/2.
c
      counter=0.
      em=0.
      dem=0.
      tbeta=0.
      dtbeta=0.
      tgamma=0.
      dtgamma=0.
      talfa=0.
      dtalfa=0.
c
      DO 100,i=1,npoints
          IF (DABS((akm-akappa(i)*1.d3)/akm).LT.1.d-2) GOTO 100
          bubu=(sigma(i)*1.d-3)**2.
          bubu=(bubu-sig2min)
          IF (bubu.LT.0) GOTO 100
          counter=counter+1.
          bubu=bubu/qdl**2.
     1        /(akappa(i)*1.d3-akm)**2.
c
          em1=DSQRT(bubu*sig2min/qdl**2.)
          em=em1+em
          dem=dem+em1**2.
c
          tbeta1=bubu/em1
          tgamma1=(bubu*(1.+akm*qdl)**2.+sig2min)/qdl**2./em1
          talfa1=bubu*(1.+akm*qdl)/qdl/em1
c
          tbeta1=tbeta1+(quadlen/2.)**2*tgamma1+quadlen*talfa1
          tgamma1=tgamma1
          talfa1=quadlen/2.*tgamma1+talfa1
c
          tbeta=tbeta+tbeta1
          dtbeta=dtbeta+tbeta1**2.
c

          tgamma=tgamma+tgamma1
          dtgamma=dtgamma+tgamma1**2.
c
          talfa=talfa+talfa1
          dtalfa=dtalfa+talfa1**2
c
100   CONTINUE
c
      em=em/counter
      dem=dem/counter
      dem=DSQRT(dem-em**2)
c
      tbeta=tbeta/counter
      dtbeta=dtbeta/counter
      dtbeta=DSQRT(dtbeta-tbeta**2)
c
      tgamma=tgamma/counter
      dtgamma=dtgamma/counter
```

```
      dtgamma=DSQRT(dtgamma-tgamma**2)
c
      talfa=talfa/counter
      dtalfa=dtalfa/counter
      dtalfa=DSQRT(dtalfa-talfa**2)
c
c     -------
      RETURN
      END
c
c     ##########################################################
      SUBROUTINE Topoutput
c
      IMPLICIT REAL*8 (A-H,O-Z)
      INCLUDE 'Comblock.for'
c
      WRITE (30,1000) title
      WRITE (30,1001) akmin*1.d3*1.1,akmax*1.d3*.9,sigmin**2/2.,
     1               2.*sigmax**2
c
      DO 100,i=1,npoints
          WRITE(30,1002) akappa(i)*1.d3,sigma(i)**2.
     1                -dsigma(i)*2*sigma(i)
          WRITE(30,1002) akappa(i)*1.d3,sigma(i)**2.
     1                +dsigma(i)*2*sigma(i)
          WRITE(30,*) 'JOIN SOLID'
100   CONTINUE
c
      DO 200,cc=1.1*akmin,.9*akmax,(.9*akmax-1.1*akmin)/30.
          yy=par(1)*cc**2+par(2)*cc+par(3)
          WRITE (30,1002) cc*1.d3,yy
200   CONTINUE
      WRITE (30,*),'JOIN SPLINE SOLID'
c
      RETURN
c
c     --------
1000  FORMAT ('TITLE TOP "',A80,'"')
1001  FORMAT ('SET LIMITS X FROM ',e9.3,' TO ',e9.3,' Y FROM ',
     1        e9.3,' TO ',e9.3)
1002  FORMAT (2x,e16.9,2x,e16.9)
c
      END
c
c     ##########################################################
      SUBROUTINE Output
c
      IMPLICIT REAL*8 (A-H,O-Z)
      CHARACTER*2 akind,strhlp(2)
      CHARACTER*5 unit,aekind
      CHARACTER*8 unit1
      INCLUDE 'Comblock.for'
c
c     ---------- File Output.dat ----------
c
      WRITE (20,*) ' Emittance 1.0  (August 92)'
      WRITE (20,1000)
      WRITE (20,1001) title
      WRITE (20,1000)
      WRITE (20,*) ' TRANSFERLINE:'
      WRITE (20,1000)
      WRITE (20,1003) quadlen
      WRITE (20,1005) driftlen
      WRITE (20,1000)
      WRITE (20,1000)
      WRITE (20,*) ' BEAM PARAMETERS:'
      WRITE (20,1000)
      WRITE (20,1002) energy
      WRITE (20,1004) current
      WRITE (20,1006) bunchlen
      IF (iraveflag.EQ.1) THEN
          WRITE (20,*) '     Emittance measurement plane     : radial'
      ELSE
          WRITE (20,*) '     Emittance measurement plane     : vertical'
      END IF
      WRITE (20,1000)
      WRITE (20,1000)
      WRITE (20,1008) npoints
      WRITE (20,1000)
      WRITE (20,1120)
```

```
      WRITE (20,1122)
      DO 200 i=1,npoints
         WRITE (20,1124) i,akappa2(i),sigma(i)*1.d-3,
     1                   dsigma(i)*1.d-3
200   CONTINUE
      WRITE (20,1000)
      WRITE (20,1000)
      WRITE (20,1000)
      WRITE (20,*) ' LEAST MEAN SQUARE PARABOLA FIT: '
      WRITE (20,1000)
      WRITE (20,*) ' Sigma**2=a*Z**2+b*Z+c    with Z=-k*sin(k*L)'
      WRITE (20,1000)
      WRITE (20,1010) par(1)*1.d-12
      WRITE (20,1011) dpar(1)*1.d-12
      WRITE (20,1000)
      WRITE (20,1012) par(2)*1.d-9
      WRITE (20,1013) dpar(2)*1.d-9
      WRITE (20,1000)
      WRITE (20,1014) par(3)*1.d-6
      WRITE (20,1015) dpar(3)*1.d-6
      WRITE (20,1000)
      WRITE (20,1000)
      WRITE (20,1016) chisquare
      WRITE (20,1000)
      WRITE (20,1018) ndegfree
      WRITE (20,1000)
      WRITE (20,1000)
      WRITE (20,*) ' MEASUREMENTS RESULTS: '
      WRITE (20,1000)
      WRITE (20,1000)
      WRITE (20,1019) em
      WRITE (20,1020) dem
      WRITE (20,1000)
      WRITE (20,*) '     Twiss parameters @ the quad begining'
      WRITE (20,1000)
      WRITE (20,1021) tbeta
      WRITE (20,1022) dtbeta
      WRITE (20,1000)
      WRITE (20,1023) tgamma
      WRITE (20,1024) dtgamma
      WRITE (20,1000)
      WRITE (20,1025) talfa
      WRITE (20,1026) dtalfa
      WRITE (20,1000)
c
c     -------
      RETURN
c
1000  FORMAT ('  ')
1001  FORMAT (1x,A80)
1002  FORMAT (5x,' Energy (MeV)                        : ',g10.4)
1003  FORMAT (5x,' Quad. length (m)                    : ',g10.4)
1004  FORMAT (5x,' Current (A)                         : ',g10.4)
1005  FORMAT (5x,' Drift length (m)                    : ',g10.4)
1006  FORMAT (5x,' Bunch Length (sec)                  : ',g10.4)
1008  FORMAT (5x,' Number of data                      : ',i4)
1010  FORMAT (5x,' a               : ',g10.4,' (m-4)')
1011  FORMAT (5x,' delta a         : ',g10.1,' (m-4)')
1012  FORMAT (5x,' b               : ',g10.4,' (m-3)')
1013  FORMAT (5x,' delta b         : ',g10.1,' (m-3)')
1014  FORMAT (5x,' c               : ',g10.4,' (m-2)')
1015  FORMAT (5x,' delta c         : ',g10.1,' (m-2)')
1016  FORMAT (5x,' Chisquare                        : ',g10.4)
1018  FORMAT (5x,' Number of degrees of freedom  : ',i5)
1019  FORMAT (5x,' Geometric emittance   : ',g10.4,' (m rad)')
1020  FORMAT (5x,' Standard deviation    : ',g10.1,' (m rad)')
1021  FORMAT (5x,' Beta                  : ',g10.4,' (m)')
1022  FORMAT (5x,' Standard deviation    : ',g10.1,' (m)')
1023  FORMAT (5x,' Gamma                 : ',g10.4,' (m-1)')
1024  FORMAT (5x,' Standard deviation    : ',g10.1,' (m-1)')
1025  FORMAT (5x,' Alpha                 : ',g10.4)
1026  FORMAT (5x,' Standard deviation    : ',g10.1)
1120  FORMAT ('            k2 (m-2)    Sigma (m)    Delta sigma (m)')
1122  FORMAT ('-----------------------------------------------------------
     1--')
1124  FORMAT (2x,i3,6x,e10.4,4x,e10.4,5x,e10.4)
c
      END
```

COMBLOCK.FOR

```
c       COMMON BLOCK for PROGRAM 'EMITTANCE'
c
        CHARACTER*80 title,blob
        CHARACTER*10 phaseplane(2)
c
        COMMON/BEAM/energy,current,bunchlen
        COMMON/LINE/quadlen,driftlen
        COMMON/TWISS/tbeta,dtbeta,tgamma,dtgamma,talfa,dtalfa,em,dem
        COMMON/FLAGS/iraveflag,izerodsigflag
        COMMON/STRINGS/title,blob,phaseplane
        COMMON/DATA/npoints,akappa(100),sigma(100),dsigma(100),akmin,
     1          akmax,sigmin,sigmax,akappa2(100)
        COMMON/GENERAL/pem,pec,pc,pi
        COMMON/HELP/gamma
        COMMON/LMS/h(3,3),a(3),par(3),dpar(3),chisquare,ndegfree
```

INPUT.DAT

```
EMITTANCE MEASUREMENT
BEAM PARAMETERS:
Energy (MeV)           Current (A)            Bunch Length (sec)
   510                    0.1                        10.d-9
-------------
-------------
      Quad. length (m)          Drift Length (m)
               .2                      2.9
-------------
-------------
Phase plane: Radial=1 Vertical=2
               1
-------------
-------------
Number of points
        10
-------------
-------------
-------------
-------------
  n        k2 (m-2)         sigma (mm)       dsigma (mm)
---------------------------------------------------------------------
  7          2.2              1.50             .1
  8          2.3              1.33             .05
  9          2.4              1.22             .13
 10          2.5              1.13             .03
 11          2.6              1.06             .07
 12          2.7              1.01             .02
 13          2.8              .996             .08
 14          2.9              1.01             .1
 15          3.0              1.05             .01
 16          3.1              1.12             .04
```

OUTPUT.DAT

```
Emittance 1.0  (August 92)

 EMITTANCE MEASUREMENT

 TRANSFERLINE:

    Quad. length (m)                 : 0.2000
    Drift length (m)                 :  2.900


 BEAM PARAMETERS:

    Energy (MeV)                     :  510.0
    Current (A)                      : 0.1000
    Bunch Length (sec)               : 0.1000E-07
    Emittance measurement plane      : radial


    Number of data                   :   10

           k2 (m-2)    Sigma (m)    Delta sigma (m)
------------------------------------------------------------
  1      0.2200E+01    0.1500E-02     0.1000E-03
  2      0.2300E+01    0.1330E-02     0.5000E-04
  3      0.2400E+01    0.1220E-02     0.1300E-03
  4      0.2500E+01    0.1130E-02     0.3000E-04
  5      0.2600E+01    0.1060E-02     0.7000E-04
  6      0.2700E+01    0.1010E-02     0.2000E-04
  7      0.2800E+01    0.9960E-03     0.8000E-04
  8      0.2900E+01    0.1010E-02     0.1000E-03
  9      0.3000E+01    0.1050E-02     0.1000E-04
 10      0.3100E+01    0.1120E-02     0.4000E-04


 LEAST MEAN SQUARE PARABOLA FIT:

 Sigma**2=a*Z**2+b*Z+c     with Z=-k*sin(k*L)

    a          : 0.8475E-04 (m-4)
    delta a    :    0.1E-04 (m-4)

    b          : 0.9334E-04 (m-3)
    delta b    :    0.1E-04 (m-3)

    c          : 0.2668E-04 (m-2)
    delta c    :    0.4E-05 (m-2)


    Chisquare                      : 0.1621

    Number of degrees of freedom   :     7

 MEASUREMENTS RESULTS:


    Geometric emittance    : 0.1034E-05 (m rad)
    Standard deviation     :    0.4E-07 (m rad)

    Twiss parameters @ the quad begining

    Beta                   :  9.054     (m)
    Standard deviation     :    0.4     (m)

    Gamma                  : 0.5528     (m-1)
    Standard deviation     :    0.2E-01 (m-1)

    Alpha                  : -2.001
    Standard deviation     :    0.9E-01
```


EMITTANCE.TOP
EMITTANCE MEASUREMENT
σ² (cm²)
4
3
2
1
-0.40
-0.45
-0.50
-0.55
-0.60
-0.65
A (quad strength) (m⁻¹)